

Lesson 22

Parametric Curves, Arc Length and Surface Area

Initializations

```
> restart;  
>
```

22.1 Arc Length and Surface Area

Formulas for the arc length of parametrically defined curves and for the surface area of a solid obtained by revolving a parametrically defined curve about a given axis, are similar to the ones we have seen before. Since

$$(ds)^2 = (dx)^2 + (dy)^2$$

the arclength of the curve

$$x = x(t) \quad y = y(t) \quad a \leq t \leq b$$

is given by

$$s = \int_a^b \sqrt{(x'(t))^2 + (y'(t))^2} dt$$

Similarly, since $dS = 2\pi r ds$, the area of the surface of revolution obtained by revolving the curve

$$x = x(t) \quad y = y(t) \geq 0 \quad a \leq t \leq b$$

about the x -axis is given by

$$S = 2\pi \int_a^b y(t) \sqrt{(x'(t))^2 + (y'(t))^2} dt$$

Examples

Example 22.1.1

Let C denote the curve

$$x(t) = e^t \cos t \quad y(t) = e^t \sin t \quad 0 \leq t \leq \pi$$

- i) Plot the curve.
- ii) Compute the arclength of this curve.
- iii) Find the area of the surface obtained by revolving this curve about the x -axis.

Solution


- i) Plot the curve.

Code the parametric equations and plot the curve.

```
> f:=t->exp(t)*cos(t);  
> g:=t->exp(t)*sin(t);  
f:=t→etcos(t)
```

$$g := t \rightarrow e^t \sin(t) \quad (2.1.1.1)$$

```
> plot([f(t), g(t), t=0..Pi], color=blue, scaling=
constrained, tickmarks=[10,5]);
```



ii) Compute the arc length of this curve.

Code and evaluate the integral that represents the arc length.

```
> e1:=Int(sqrt(D(f)(t)^2+D(g)(t)^2), t=0..Pi);
```

$$e1 := \int_0^{\pi} \sqrt{(e^t \cos(t) - e^t \sin(t))^2 + (e^t \sin(t) + e^t \cos(t))^2} dt \quad (2.1.1.2)$$

```
> e2:=simplify(e1);
```

$$e2 := \sqrt{2} \left(\int_0^{\pi} e^t dt \right) \quad (2.1.1.3)$$

The arc length L of this curve equals

```
> L:=value(e2);
evalf(L);
```

$$L := \sqrt{2} (-1 + e^{\pi}) \quad (2.1.1.4)$$

31.31166780

iii) Find the area of the surface obtained by revolving this curve about the x -axis.

Code and evaluate the integral that represents the surface area.

```
> a1:=Int(2*Pi*g(t)*sqrt(D(f)(t)^2+D(g)(t)^2), t=0..Pi);
```

$$a1 := \int_0^{\pi} 2 \pi e^t \sin(t) \sqrt{(e^t \cos(t) - e^t \sin(t))^2 + (e^t \sin(t) + e^t \cos(t))^2} dt \quad (2.1.1.5)$$

```
> a2:=simplify(a1);
```

$$(2.1.1.6)$$

$$a2 := 2 \pi \sqrt{2} \left(\int_0^{\pi} e^{2t} \sin(t) dt \right) \quad (2.1.1.6)$$

The surface area S equals


```
> S:=simplify(value(a2));
evalf(S);
```

$$S := \frac{2}{5} \pi \sqrt{2} (1 + e^{2\pi})$$

953.4278500 (2.1.1.7)

It is possible to plot this solid of revolution. The mathematics behind the code will be supplied in Calculus III. The result is shown below.

```
> plot3d([f(t), g(t)*cos(u), g(t)*sin(u)], t=0..Pi, u=0..2*
Pi, style=patch, orientation=[-52, 55], axes=boxed,
scaling=constrained, labels=[x, y, z], tickmarks=[10, 10,
10]);
```



>